# $Id: PKGBUILD 67388 2012-03-10 21:54:03Z lcarlier $
# Maintainer : Geoffroy Carrier <gcarrier@aur.archlinux.org>
# Contributor: Joel Schaerer <joel.schaerer@laposte.net>

pkgname=slimevolley
pkgver=2.4.2
pkgrel=2
pkgdesc="A simple yet fun arcade volley game"
arch=('i686' 'x86_64')
url="http://slime.tuxfamily.org/index.php"
license=('GPL')
depends=('sdl_net' 'sdl_ttf' 'sdl_image')
makedepends=('cmake')
source=("http://downloads.tuxfamily.org/slime/v242/${pkgname}_${pkgver}.tar.bz2"
        link-libm.patch)
md5sums=('42d1287bf459c26475a9bb7a9bb167e5')

build() {
  cd "$srcdir/$pkgname"

  patch -Np1 -i ../link-libm.patch

  cmake -DCMAKE_INSTALL_PREFIX:FILEPATH=/usr .
  make
}

package() {
  cd "$srcdir/$pkgname"

  make DESTDIR="${pkgdir}" install
}
md5sums=('42d1287bf459c26475a9bb7a9bb167e5'
         '4e5a7d4fb97e8f482e5468acc795a17b')
